PharmaSUG 2016 - Paper SP05

Unequalslopes: Making life easier when the proportional odds assumption fails

Matthew Wiedel, inVentiv Health, Lincoln, NE

ABSTRACT

With the advent of SAS 9.3 came the proc logistic model option, unequalslopes. This new option allows the programmer to quickly produce results for cumulative logit models which fail the assumption of proportionality. Models thus become either partially proportional or non-proportional. Previously, these models could be programmed using the procedure NLMixed, but this new capability of proc logistic provides an easier task for the SAS writer and an analyst friendly output. Ordinal responses use these models to capture the inherent information of their ordering. Hoping to make it easy to generalize the interpretation and programming of partial and non-proportional logit equations, the response variable, number of episodes, will be a count (1,2,3, or 4) against 2 categorical independent variables, rank and level, with 5 and 3 categories, respectively. Working from simpler to more complex models, the body of the paper will be split into three parts: Proportional Odds Model, Partial Proportional Odds Model, and Non-Proportional Odds Model. Each section will display the model equations, the NLmixed code, NLmixed results logistic code, and finally logistic results. It would be easy to digress, thus the focus will be on looking at the proportional odds assumption and interpretation of the odds ratios. The NLmixed code along with the mathematical equations should give a deeper understanding of the logistic output, hence making it easier to explain the results to any audience.

INTRODUCTION

Often the request for exploratory statistics may require a programmer to write code and interpret output which he or she hasn't had the opportunity to do in the past. There are times when the programming is simple and the results are easily comprehended. However, there are times when there isn't a procedure which readily produces the sought for measures and the output isn't as easily understood by all parties. The latter situation was the inspiration for this *paper.*

Cumulative Logit Models are often used to make inferences about ordinal response data. They also can come in many forms, but the three which will be discussed are the Proportional Odds Model, Partial Proportional Odds Model, and the Non-Proportional Odds Model. The models increase in complexity as the assumptions of proportionality among any or all of the independent variables become false.

Two SAS procedures, NLMixed and Logistic, will be used to compare results. NLMixed is much more coding intensive than using the options in Logistic. In SAS 9.3, the Logistic procedure added the model option, unequalslopes, to address partial or non-proportionality among the explanatory categories in the logit model.

Cumulative Proportional Odds Logistic Model

Logistic regression is the categorical sister to linear regression. Linear regression deals with continuous dependent variables and continuous or binary independent variables. Logistic regression tackles problems where the outcome is a finite collection of predetermined groups modeled by explanatory values which are either continuous or categorical.

When dealing with categorical outcomes, the models look at the odds of those outcomes: p/1-p, p=probability of outcome. A logit is the natural log of an odds, ln(p/1-p), and it is often the transformation used when modeling categorical outcomes. The dependent variable may be just a collection of different categorical results or may have an ordering of severity, number, or rank. In these cases, a cumulative logistic model is used where one compares the probabilities of occurrences at and below a level compared to all the higher level occurrences: $\frac{P[Y \leq Sr]}{P[Y > Sr]}$, r= # of response levels However, as in the case here, the inequality signs are flipped in order to talk about the odds ratios differently: $\frac{P[Y > Sr]}{P[Y \leq Sr]}$ The rest of the model will have the same form $\alpha + \beta X$ $\alpha$is the intercept, β is the coefficient vector, and X is the independent variable matrix.

With this as a starting point, one looks at the data and the questions asked. Here, the response is the number of episodes (1, 2, 3, 4) and the two explanatory categories are rank (5 groups) and level (3 groups). Since the explanatory variables are "ranks" and "levels", they will have an inherent ordering also. In an attempt to capture this ordering among the independent variables, the lowest rank and level were used as reference groups. All subjects will have at least one episode.

An assumption is that the lower groups will have a smaller number of episodes than the higher groups.

Along with this is another thought that the further the groups are apart, the greater in disparity of number of episodes. Although the response is a collection of integers, the measurements will be in probabilities and odds where the discussions are in the chances of this group having r or more episodes as opposed to this other group. The hermeneutics will have to be shared by all parties.

Given all this information, the researcher asked for the standard fare of counts, odds ratios, confidence intervals, and p-values. A cumulative logit model with an assumption of proportional odds is initially used, and the logistic procedure handles this beautifully. As for every model, a test of fit is done and typically a p-value is produced. The assumption of proportional odds does a chi-squared test on the β's to check this assumption:

$L_1 = \alpha_1 + \beta_{11}$ x rank + β_{21} x level {$L_1 = \ln(\frac{P[\#\text{ of episodes}>3]}{P[\#\text{ of episodes}\leq 3]})$ 4 vs. 1, 2 or 3}

$L_2 = \alpha_2 + \beta_{12}$ x rank + β_{22} x level {$L_2 = \ln(\frac{P[\#\text{ of episodes}>2]}{P[\#\text{ of episodes}\leq 3]})$ 3 or 4 vs. 1 or 2}

$L_3 = \alpha_3 + \beta_{13}$ x rank + β_{23} x level {$L_3 = \ln(\frac{P[\#\text{ of episodes}>1]}{P[\#\text{ of episodes}\leq 1]})$ 2 or 3 or 4 vs. 1}

Ho: $\beta_{11} = \beta_{12} = \beta_{13}$ and $\beta_{21} = \beta_{22} = \beta_{23}$

If the null hypothesis holds true, then the assumption of proportional odds is confirmed and the models can be written as such:

$L_1 = \alpha_1 + \beta_1$ x rank + β_2 x level {$L_1 = \ln(\frac{P[\#\text{ of episodes}>3]}{P[\#\text{ of episodes}\leq 3]})$ 4 vs. 1, 2 or 3}

$L_2 = \alpha_2 + \beta_1$ x rank + β_2 x level {$L_2 = \ln(\frac{P[\#\text{ of episodes}>2]}{P[\#\text{ of episodes}\leq 3]})$ 3 or 4 vs. 1 or 2}

$L_3 = \alpha_3 + \beta_1$ x rank + β_2 x level {$L_3 = \ln(\frac{P[\#\text{ of episodes}>1]}{P[\#\text{ of episodes}\leq 1]})$ 2 or 3 or 4 vs. 1}

For our example, the lowest ranked category and lowest level as references. Although the statistics sought are easily produced from the logistic procedure, to help illustrate the evolution of the nlmixed code as the models become more complex, both approaches are given. The NLmixed procedure does outline in it's code how the statistics are produced.

There are a lot of things that are going on in the nlmixed code that are both intuitive and not intuitive. After initiating the intercept parameters and setting up the coefficient definitions, there are four cumulative probability (cp) equation which can be interpreted as:

cp4 = P(# of episodes >= 4|rank=(2, 3, 4, or 5) and level=(2 or 3))

cp3 = P(# of episodes >= 3|rank=(2, 3, 4, or 5) and level=(2 or 3))

cp2 = P(# of episodes >= 2|rank=(2, 3, 4, or 5) and level=(2 or 3))

cp1 = P(# of episodes >= 1|rank=(2, 3, 4, or 5) and level=(2 or 3))

Following this are the definitions of individual probabilities (ip):

ip4 = P(# of episodes = 4|rank=(2, 3, 4, or 5) and level=(2 or 3))

ip3 = P(# of episodes = 3|rank=(2, 3, 4, or 5) and level=(2 or 3))

ip2 = P(# of episodes = 2|rank=(2, 3, 4, or 5) and level=(2 or 3))

ip1 = P(# of episodes = 1|rank=(2, 3, 4, or 5) and level=(2 or 3))

In a normal analysis, the order in which the individual probabilities would go from 1 to 4 instead of 4 to 1. This is the place where the descending order and the logit's signs flip. The programming of the cumulative probabilities would be written from cp1 to cp4 to show the how the probabilities would be treated .

Then the model statement is created using the log likelihood value (p) for each individual probability. At this point, nlmixed will output coefficient estimates and some diagnostics. From those coefficients one can manually compute the odds ratios. However, in order to replicate the logistic output, estimate statements were used. Below is what is produced.

```
proc nlmixed data=rank_level_only;
 parms Intercept4 = -1, Intercept3 = 0, Intercept2 = 1, Intercept1 = 2;

 r2=(rank=2); r3=(rank=3); r4=(rank=4); r5=(rank=5);
 l2=(level=2);l3=(level=3);
```

```
cp4= 1/(1 + exp(-(Intercept4 + rank2*r2 + rank3*r3 + rank4*r4 + rank5*r5 + level2*l2 + level3*l3)));
cp3= 1/(1 + exp(-(Intercept3 + rank2*r2 + rank3*r3 + rank4*r4 + rank5*r5 + level2*l2 + level3*l3)));
cp2= 1/(1 + exp(-(Intercept2 + rank2*r2 + rank3*r3 + rank4*r4 + rank5*r5 + level2*l2 + level3*l3)));
cp1= 1/(1 + exp(-(Intercept1 + rank2*r2 + rank3*r3 + rank4*r4 + rank5*r5 + level2*l2 + level3*l3)));

if      episodes=4 then ip = cp4;
else if episodes=3 then ip = cp3-cp4;
else if episodes=2 then ip = cp2-cp3;
else if episodes=1 then ip = cp1-cp2;
else                        ip = 1-cp1;

p = (ip>0 and ip<=1)*ip + (ip<=0)*1e-8 + (ip>1);
loglik = log(p);
model episodes ~ general(loglik);
replicate count;
id cp1-cp4;
predict ip out=nlm;
estimate "rank 2 vs. 1" exp(rank2);
estimate "rank 3 vs. 1" exp(rank3);
estimate "rank 4 vs. 1" exp(rank4);
estimate "rank 5 vs. 1" exp(rank5);
estimate "level 2 vs. 1" exp(level2);
estimate "level 3 vs. 1" exp(level3);
run;
```

The NLMIXED Procedure

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper	Gradient
Intercept4	-1.6367	0.1999	60	-8.19	<.0001	0.05	-2.0366	-1.2368	0.006324
Intercept3	-0.08262	0.1934	60	-0.43	0.6707	0.05	-0.4695	0.3042	-0.00213
Intercept2	1.5597	0.2050	60	7.61	<.0001	0.05	1.1497	1.9696	0.001691
Intercept1	18.3747	390.65	60	0.05	0.9626	0.05	-763.04	799.79	-6.55E-6
rank2	0.3258	0.1932	60	1.69	0.0970	0.05	-0.06078	0.7123	-0.00113
rank3	0.5648	0.1921	60	2.94	0.0047	0.05	0.1805	0.9492	-0.00056
rank4	0.3335	0.2128	60	1.57	0.1224	0.05	-0.09220	0.7592	-0.00119
rank5	0.5177	0.2910	60	1.78	0.0803	0.05	-0.06435	1.0998	0.001023
level2	-0.06254	0.1568	60	-0.40	0.6914	0.05	-0.3762	0.2511	-0.00017
level3	0.1964	0.1614	60	1.22	0.2284	0.05	-0.1264	0.5192	0.000621

Additional Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
rank 2 vs. 1	1.3851	0.2677	60	5.17	<.0001	0.05	0.8497	1.9205
rank 3 vs. 1	1.7592	0.3380	60	5.20	<.0001	0.05	1.0830	2.4353
rank 4 vs. 1	1.3958	0.2970	60	4.70	<.0001	0.05	0.8016	1.9900
rank 5 vs. 1	1.6782	0.4883	60	3.44	0.0011	0.05	0.7014	2.6550
level 2 vs. 1	0.9394	0.1473	60	6.38	<.0001	0.05	0.6447	1.2340
level 3 vs. 1	1.2170	0.1964	60	6.20	<.0001	0.05	0.8241	1.6099

The succinct logistic procedure will create the same model coefficient and odds ratio estimates as using nlmixed (slight rounding differences).

```
proc logistic data=rank_level_only descending;
class  rank (ref='1') level (ref='1')/ param=ref;
model episodes = rank level;
freq count;
run;
```

Initially, we are given the design variables. These numbers help with estimate statements and make the calculation of odds ratios simple since the reference groups (1) are all zeros. As indicated in the NLmixed estimate statements, all one has to do is take the exponent of coefficients to get the odds ratios.

Class Level Information

Class Value Design Variables

rank	1	0	0	0	0
	2	1	0	0	0
	3	0	1	0	0
	4	0	0	1	0
	5	0	0	0	1
level	1	0	0		
	2	1	0		
	3	0	1		

Score Test for the Proportional Odds Assumption

Chi-Square	DF	Pr > ChiSq
39.8669	12	**<.0001**

Above is the diagnostic that indicates that the model is not adequate. Among the different rank groups or different level groups or both, the coefficients are not the same for each cumulative odds ratio.

Analysis of Maximum Likelihood Estimates

Parameter		DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept 4		1	-1.6370	0.1990	67.6468	<.0001
Intercept 3		1	-0.0829	0.1912	0.1879	0.6646
Intercept 2		1	1.5594	0.2018	59.6884	<.0001
rank	2	1	0.3260	0.1959	2.7704	0.0960
rank	3	1	0.5651	0.1933	8.5451	0.0035
rank	4	1	0.3337	0.2145	2.4212	0.1197
rank	5	1	0.5178	0.2813	3.3881	0.0657
level	2	1	-0.0625	0.1555	0.1615	0.6878
level	3	1	0.1965	0.1631	1.4510	0.2284

Odds Ratio Estimates

Effect	Point Estimate	95% Wald Confidence Limits	
rank 2 vs 1	1.385	0.944	2.034
rank 3 vs 1	1.760	1.205	2.570
rank 4 vs 1	1.396	0.917	2.126
rank 5 vs 1	1.678	0.967	2.913
level 2 vs 1	0.939	0.693	1.274
level 3 vs 1	1.217	0.884	1.676

Let OR_j = (Odds of # of episodes >= j:Odds of # of episodes < j). When the odds are proportional, the OR_j is the same for each explanatory variable comparison across the j's (i.e. for rank 4 vs 1, $OR_4 = OR_3 = OR_2 = e^{.3337} = 1.396$). This notation makes it easier to understand when the assumption of proportionality fails which is the case here.

There are a couple things to note about the logistic output vs. the NLmixed output. The logistic procedure uses chi-squared tests to get p-values and confidence intervals as opposed with NLmixed which uses t-values. Also, if we ignore the proportionality assumption, there are two ways to tell which relationship is significant, the p-value of the coefficient or confidence limits of the odds ratio. Here, rank 3 vs 1 is the only significant relationship, the p-value = .0035 and 1 is not in the OR Confidence Interval (1.205, 2.570) (if a ratio = 1, then the numerator and denominator are equal).

To address the proportionality assumption failing, the number of episodes was modeled against each explanatory variable alone to see which is causing the lack of proportionality.

$L_1 = \alpha_1 + \beta_{11} \times \text{rank}$ $\{L_1 = \ln\left(\frac{P[\text{\# of episodes}>3]}{P[\text{\# of episodes}\leq 3]}\right)$ 4 vs. 1, 2 or 3}

$L_2 = \alpha_2 + \beta_{12}$ x rank {$L_2 = \ln(\frac{P[\#\text{ of episodes}>2]}{P[\#\text{ of episodes}\leq 3]})$ 3 or 4 vs. 1 or 2}

$L_3 = \alpha_3 + \beta_{13}$ x rank {$L_3 = \ln(\frac{P[\#\text{ of episodes}>1]}{P[\#\text{ of episodes}\leq 1]})$ 2 or 3 or 4 vs. 1}

Ho: $\beta_{11} = \beta_{12} = \beta_{13}$

```
proc logistic data=rank_level_only descending;
class  rank (ref='1') level (ref='1')/ param=ref;
model episodes = rank;
freq count;
run;
```

Score Test for the Proportional Odds Assumption

Chi-Square	DF	Pr > ChiSq
19.2835	8	**0.0134**

$L_1 = \alpha_1 + \beta_{21}$ x level {$L_1 = \ln(\frac{P[\#\text{ of episodes}>3]}{P[\#\text{ of episodes}\leq 3]})$ 4 vs. 1, 2 or 3}

$L_2 = \alpha_2 + \beta_{22}$ x level {$L_2 = \ln(\frac{P[\#\text{ of episodes}>2]}{P[\#\text{ of episodes}\leq 3]})$ 3 or 4 vs. 1 or 2}

$L_3 = \alpha_3 + \beta_{23}$ x level {$L_3 = \ln(\frac{P[\#\text{ of episodes}>1]}{P[\#\text{ of episodes}\leq 1]})$ 2 or 3 or 4 vs. 1}

Ho: $\beta_{21} = \beta_{22} = \beta_{23}$

```
proc logistic data=rank_level_only descending;
class  rank (ref='1') level (ref='1')/ param=ref;
model episodes = level;
freq count;
run;
```

Score Test for the Proportional Odds Assumption

Chi-Square	DF	Pr > ChiSq
23.9110	4	**<.0001**

It appears as though both rank and level are causing the model to be non-proportional. Hence, a non-proportional odds model would be sought, and that's when we get to use the new logistic option of 'unequalslopes'. For purposes of illustration, a partial proportional odds model will be created assuming proportionality for the level variables.

Cumulative Partial Proportional Odds Logit Model:

$\ln(\frac{P[Y>Sr]}{P[Y\leq Sr]}) = \alpha_r + \beta_{1r}\text{rank} + \beta_2\text{level}$

$L_1 = \alpha_1 + \beta_{11}$ x rank + β_{21} x level {$L_1 = \ln(\frac{P[\#\text{ of episodes}>3]}{P[\#\text{ of episodes}\leq 3]})$ 4 vs. 1, 2 or 3}

$L_2 = \alpha_2 + \beta_{12}$ x rank + β_{22} x level {$L_2 = \ln(\frac{P[\#\text{ of episodes}>2]}{P[\#\text{ of episodes}\leq 3]})$ 3 or 4 vs. 1 or 2}

$L_3 = \alpha_3 + \beta_{13}$ x rank + β_{23} x level {$L_3 = \ln(\frac{P[\#\text{ of episodes}>1]}{P[\#\text{ of episodes}\leq 1]})$ 2 or 3 or 4 vs. 1}

Ho: $\beta_{11} = \beta_{12} = \beta_{13}$ (failed)

Ho: $\beta_{21} = \beta_{22} = \beta_{23}$ (Pretending did not fail)

```
proc nlmixed data=rank_level_only;
 parms Intercept4 = -1, Intercept3 = 0, Intercept2 = 1, Intercept1 = 2;

 r2=(rank=2); r3=(rank=3); r4=(rank=4); r5=(rank=5);
 l2=(level=2);l3=(level=3);

 cp4= 1/(1 + exp(-(Intercept4 + rank24*r2 + rank34*r3 + rank44*r4 + rank54*r5 + level2*l2 + level3*l3)));
 cp3= 1/(1 + exp(-(Intercept3 + rank23*r2 + rank33*r3 + rank43*r4 + rank53*r5 + level2*l2 + level3*l3)));
 cp2= 1/(1 + exp(-(Intercept2 + rank22*r2 + rank32*r3 + rank42*r4 + rank52*r5 + level2*l2 + level3*l3)));
 cp1= 1/(1 + exp(-(Intercept1 + rank21*r2 + rank31*r3 + rank41*r4 + rank51*r5 + level2*l2 + level3*l3)));

 if      episodes=4 then ip = cp4;
 else if episodes=3 then ip = cp3-cp4;
 else if episodes=2 then ip = cp2-cp3;
 else if episodes=1 then ip = cp1-cp2;
 else                    ip = 1-cp1;

 p = (ip>0 and ip<=1)*ip + (ip<=0)*1e-8 + (ip>1);
 loglik = log(p);
 model episodes ~ general(loglik);
 replicate count;
 id cp1-cp4;
 predict ip out=nlm;
 estimate "rank 2 vs. 1  4" exp(rank24);
 estimate "rank 2 vs. 1  3" exp(rank23);
 estimate "rank 2 vs. 1  2" exp(rank22);
 estimate "rank 3 vs. 1  4" exp(rank34);
 estimate "rank 3 vs. 1  3" exp(rank33);
 estimate "rank 3 vs. 1  2" exp(rank32);
 estimate "rank 4 vs. 1  4" exp(rank44);
 estimate "rank 4 vs. 1  3" exp(rank43);
 estimate "rank 4 vs. 1  2" exp(rank42);
 estimate "rank 5 vs. 1  4" exp(rank54);
 estimate "rank 5 vs. 1  3" exp(rank53);
 estimate "rank 5 vs. 1  2" exp(rank52);
 estimate "level 2 vs. 1   " exp(level2);
 estimate "level 3 vs. 1   " exp(level3);
run;
```

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper	Gradient
Intercept4	-1.9978	0.2889	60	-6.92	<.0001	0.05	-2.5757	-1.4200	0.000269
Intercept3	0.1541	0.2077	60	0.74	0.4609	0.05	-0.2613	0.5695	-0.00022
Intercept2	1.3654	0.2446	60	5.58	<.0001	0.05	0.8761	1.8547	-0.00029
Intercept1	18.0323	743.61	60	0.02	0.9807	0.05	-1469.41	1505.48	-2.06E-6
rank24	0.5745	0.3082	60	1.86	0.0672	0.05	-0.04201	1.1910	0.000127
rank34	0.9982	0.2979	60	3.35	0.0014	0.05	0.4024	1.5940	-0.00004
rank44	0.7742	0.3246	60	2.38	0.0203	0.05	0.1248	1.4235	0.000126
rank54	1.2221	0.3818	60	3.20	0.0022	0.05	0.4583	1.9859	0.000026
level2	-0.06973	0.1567	60	-0.45	0.6579	0.05	-0.3831	0.2437	0.000041
level3	0.1898	0.1617	60	1.17	0.2451	0.05	-0.1336	0.5133	0.000084
rank23	0.1121	0.2175	60	0.52	0.6082	0.05	-0.3230	0.5472	0.000076
rank33	0.2890	0.2151	60	1.34	0.1842	0.05	-0.1413	0.7192	-0.00013
rank43	0.004334	0.2377	60	0.02	0.9855	0.05	-0.4712	0.4798	-0.00012
rank53	0.1414	0.3137	60	0.45	0.6538	0.05	-0.4861	0.7690	-0.00022
rank22	0.7136	0.2981	60	2.39	0.0198	0.05	0.1174	1.3099	-0.00004
rank32	0.7182	0.2918	60	2.46	0.0167	0.05	0.1345	1.3019	-0.00016
rank42	0.7157	0.3356	60	2.13	0.0371	0.05	0.04442	1.3871	-0.00003
rank52	0.2636	0.4115	60	0.64	0.5243	0.05	-0.5596	1.0867	-0.00002
rank21	4.4571	4786.40	60	0.00	0.9993	0.05	-9569.77	9578.68	-4.47E-8
rank31	4.6148	4914.22	60	0.00	0.9993	0.05	-9825.28	9834.51	-4.24E-8
rank41	3.4240	3655.91	60	0.00	0.9993	0.05	-7309.48	7316.33	-7.8E-8
rank51	2.3100	3431.36	60	0.00	0.9995	0.05	-6861.43	6866.05	-8.91E-8

Additional Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
rank 2 vs. 1 4	1.7762	0.5474	60	3.24	0.0019	0.05	0.6812	2.8712
rank 2 vs. 1 3	1.1186	0.2433	60	4.60	<.0001	0.05	0.6319	1.6053
rank 2 vs. 1 2	2.0414	0.6085	60	3.35	0.0014	0.05	0.8242	3.2586
rank 3 vs. 1 4	2.7134	0.8082	60	3.36	0.0014	0.05	1.0967	4.3302
rank 3 vs. 1 3	1.3350	0.2871	60	4.65	<.0001	0.05	0.7607	1.9094
rank 3 vs. 1 2	2.0508	0.5984	60	3.43	0.0011	0.05	0.8537	3.2479
rank 4 vs. 1 4	2.1688	0.7041	60	3.08	0.0031	0.05	0.7604	3.5771
rank 4 vs. 1 3	1.0043	0.2388	60	4.21	<.0001	0.05	0.5268	1.4819
rank 4 vs. 1 2	2.0457	0.6866	60	2.98	0.0042	0.05	0.6724	3.4190
rank 5 vs. 1 4	3.3943	1.2960	60	2.62	0.0112	0.05	0.8018	5.9867
rank 5 vs. 1 3	1.1519	0.3614	60	3.19	0.0023	0.05	0.4290	1.8748
rank 5 vs. 1 2	1.3016	0.5356	60	2.43	0.0181	0.05	0.2302	2.3730
level 2 vs. 1	0.9326	0.1461	60	6.38	<.0001	0.05	0.6403	1.2250
level 3 vs. 1	1.2090	0.1955	60	6.18	<.0001	0.05	0.8180	1.6001

Now the analysis has gotten a bit more complicated, and it is important to be able to communicate what is actually being measured. For level, it is as before: the odds when comparing level 2 with 1 is the same for each cumulative OR *4 episodes vs. 3, 2, or 1, 4 or 3 episodes vs. 2 or 1, 4 or 3 or 2 episodes vs. 1.* However, for rank, the odds ratios become specific to symptom comparison. One can see that this will be the case since the β's are now all different for each probability equation.

Comparing the odds of rank = 3 vs 1 having 4 episodes as opposed to less than 4 is different than having 4 or 3 episodes vs. 2 or 1. One could say that rank = 3 has 1.335 greater odds of having 4 or 3 episodes than rank = 1. For rank = 3 has 2.71 greater odds of having 4 episodes than rank = 1.

```
proc logistic data=rank_level_only descending;
class  rank (ref='1') level (ref='1')/ param=ref;
model episodes = rank level/unequalslopes=rank;
freq count;
run;
```

Analysis of Maximum Likelihood Estimates

Parameter		episodes	DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept		4	1	-1.9978	0.2889	47.8303	<.0001
Intercept		3	1	0.1541	0.2077	0.5509	0.4580
Intercept		2	1	1.3654	0.2446	31.1530	<.0001
rank	2	4	1	0.5745	0.3082	3.4745	0.0623
rank	2	3	1	0.1121	0.2175	0.2656	0.6063
rank	2	2	1	0.7136	0.2981	5.7316	0.0167
rank	3	4	1	0.9982	0.2979	11.2308	0.0008
rank	3	3	1	0.2890	0.2151	1.8049	0.1791
rank	3	2	1	0.7182	0.2918	6.0581	0.0138
rank	4	4	1	0.7742	0.3246	5.6868	0.0171
rank	4	3	1	0.00434	0.2377	0.0003	0.9854
rank	4	2	1	0.7157	0.3356	4.5482	0.0330
rank	5	4	1	1.2221	0.3818	10.2437	0.0014
rank	5	3	1	0.1414	0.3137	0.2033	0.6521
rank	5	2	1	0.2636	0.4115	0.4103	0.5218
level	2		1	-0.0697	0.1567	0.1981	0.6563
level	3		1	0.1898	0.1617	1.3781	0.2404

Odds Ratio Estimates

Effect	episodes	Point Estimate	95% Wald Confidence Limits	
rank 2 vs 1	4	1.776	0.971	3.250
rank 2 vs 1	3	1.119	0.730	1.713
rank 2 vs 1	2	2.041	1.138	3.662
rank 3 vs 1	4	2.713	1.513	4.865
rank 3 vs 1	3	1.335	0.876	2.035
rank 3 vs 1	2	2.051	1.158	3.633
rank 4 vs 1	4	2.169	1.148	4.098
rank 4 vs 1	3	1.004	0.630	1.600
rank 4 vs 1	2	2.046	1.060	3.949
rank 5 vs 1	4	3.394	1.606	7.174
rank 5 vs 1	3	1.152	0.623	2.131
rank 5 vs 1	2	1.302	0.581	2.916
level 2 vs 1		0.933	0.686	1.268
level 3 vs 1		1.209	0.881	1.660

This model is considered a partially proportional since the level groups are proportional, but the rank groups are not. One must be careful when interpreting these results. For the odds ratio portion of the rank groups, the episodes represents a cumulative relationship: 4 – P(4 episodes)/P(3, 2, 1 episodes); 3 – P(4 or 3 episodes)/P(2 or 1 episodes); 2 – P(4, 3, or 2 episodes)/P(1 episode). The odds ratio for the rank comparisons is as such [Odds for rank group (2, 3, 4, or 5)/Odds for rank group 1].

One thing to notice is if the confidence interval does not include 1, then the corresponding p-value for the coefficient (in essence the odds ratio itself) is <= .05 which is significant, i.e. rank= 2 vs 1 2 has a CI = (1.138, 3.662) , p-value = .0167 This makes sense since for any ratio a/b, if the possibility of a/b=1 then a=b. No proportionality test was given to test level's proportionality.

Cumulative Non-Proportional Odds Logit Model:

$\ln(\frac{P[Y>Sr]}{P[Y\leq Sr]}) = \alpha_r + \beta_{1r}\text{rank} + \beta_{2r}\text{level}$

A fully non-proportional model considers every explanatory variable non-proportional and the most complex model is formed. Each cumulative relationship for the dependent variable given each explanatory variables relationship with the reference group is given a different coefficient and odds ratio:

$$L_1 = \alpha_1 + \beta_{11} \times \text{rank} + \beta_{21} \times \text{level} \quad \{L_1 = \ln(\frac{P[\#\text{ of episodes}>3]}{P[\#\text{ of episodes}\leq 3]}) \text{ 4 vs. 1, 2 or 3}\}$$

$$L_2 = \alpha_2 + \beta_{12} \times \text{rank} + \beta_{22} \times \text{level} \quad \{L_2 = \ln(\frac{P[\#\text{ of episodes}>2]}{P[\#\text{ of episodes}\leq 2]}) \text{ 3 or 4 vs. 1 or 2}\}$$

$$L_3 = \alpha_3 + \beta_{13} \times \text{rank} + \beta_{23} \times \text{level} \quad \{L_3 = \ln(\frac{P[\#\text{ of episodes}>1]}{P[\#\text{ of episodes}\leq 1]}) \text{ 2 or 3 or 4 vs. 1}\}$$

```
proc nlmixed data=rank_level_only;
 parms Intercept4 = -1, Intercept3 = 0, Intercept2 = 1, Intercept1 = 2;

 r2=(rank=2); r3=(rank=3); r4=(rank=4); r5=(rank=5);
 l2=(level=2);l3=(level=3);

 cp4= 1/(1 + exp(-(Intercept4 + rank24*r2 + rank34*r3 + rank44*r4 + rank54*r5 + level24*l2 +
level34*l3)));
 cp3= 1/(1 + exp(-(Intercept3 + rank23*r2 + rank33*r3 + rank43*r4 + rank53*r5 + level23*l2 + level33*l3)));
 cp2= 1/(1 + exp(-(Intercept2 + rank22*r2 + rank32*r3 + rank42*r4 + rank52*r5 + level22*l2 + level32*l3)));
 cp1= 1/(1 + exp(-(Intercept1 + rank21*r2 + rank31*r3 + rank41*r4 + rank51*r5 + level21*l2 + level31*l3)));

 if      episodes=4 then ip = cp4;
 else if episodes=3 then ip = cp3-cp4;
 else if episodes=2 then ip = cp2-cp3;
 else if episodes=1 then ip = cp1-cp2;
 else                    ip = 1-cp1;
```

```
 p = (ip>0 and ip<=1)*ip + (ip<=0)*1e-8 + (ip>1);
 loglik = log(p);
 model episodes ~ general(loglik);
 replicate count;
 id cp1-cp4;
 predict ip out=nlm;
 estimate "rank 2 vs. 1  4" exp(rank24);
 estimate "rank 2 vs. 1  3" exp(rank23);
 estimate "rank 2 vs. 1  2" exp(rank22);
 estimate "rank 3 vs. 1  4" exp(rank34);
 estimate "rank 3 vs. 1  3" exp(rank33);
 estimate "rank 3 vs. 1  2" exp(rank32);
 estimate "rank 4 vs. 1  4" exp(rank44);
 estimate "rank 4 vs. 1  3" exp(rank43);
 estimate "rank 4 vs. 1  2" exp(rank42);
 estimate "rank 5 vs. 1  4" exp(rank54);
 estimate "rank 5 vs. 1  3" exp(rank53);
 estimate "rank 5 vs. 1  2" exp(rank52);
 estimate "level 2 vs. 1  4" exp(level24);
 estimate "level 2 vs. 1  3" exp(level23);
 estimate "level 2 vs. 1  3" exp(level22);
 estimate "level 3 vs. 1  4" exp(level34);
 estimate "level 3 vs. 1  3" exp(level33);
 estimate "level 3 vs. 1  2" exp(level32);
run;
```

The NLMIXED Procedure

Additional Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
rank 2 vs. 1 4	1.7118	0.5276	60	3.24	0.0019	0.05	0.6564	2.7671
rank 2 vs. 1 3	1.1363	0.2474	60	4.59	<.0001	0.05	0.6415	1.6311
rank 2 vs. 1 2	1.8810	0.5637	60	3.34	0.0015	0.05	0.7535	3.0085
rank 3 vs. 1 4	2.5728	0.7678	60	3.35	0.0014	0.05	1.0370	4.1087
rank 3 vs. 1 3	1.3566	0.2920	60	4.65	<.0001	0.05	0.7725	1.9407
rank 3 vs. 1 2	1.8125	0.5318	60	3.41	0.0012	0.05	0.7488	2.8762
rank 4 vs. 1 4	2.1049	0.6828	60	3.08	0.0031	0.05	0.7390	3.4707
rank 4 vs. 1 3	1.0245	0.2438	60	4.20	<.0001	0.05	0.5369	1.5121
rank 4 vs. 1 2	1.9599	0.6621	60	2.96	0.0044	0.05	0.6356	3.2843
rank 5 vs. 1 4	3.3600	1.2805	60	2.62	0.0110	0.05	0.7986	5.9215
rank 5 vs. 1 3	1.1572	0.3629	60	3.19	0.0023	0.05	0.4314	1.8830
rank 5 vs. 1 2	1.3484	0.5548	60	2.43	0.0181	0.05	0.2386	2.4581
level 2 vs. 1 4	1.1424	0.2334	60	4.89	<.0001	0.05	0.6755	1.6093
level 2 vs. 1 3	0.8186	0.1420	60	5.77	<.0001	0.05	0.5346	1.1026
level 2 vs. 1 3	0.9644	0.2264	60	4.26	<.0001	0.05	0.5114	1.4173
level 3 vs. 1 4	1.0363	0.2262	60	4.58	<.0001	0.05	0.5838	1.4887
level 3 vs. 1 3	1.1088	0.2036	60	5.45	<.0001	0.05	0.7015	1.5161
level 3 vs. 1 2	3.0725	0.9696	60	3.17	0.0024	0.05	1.1329	5.0120

```
proc logistic data=rank_level_only descending;
class  rank (ref='1') level (ref='1')/ param=ref;
model episodes = rank level/unequalslopes;
freq count;
run;
```

The LOGISTIC Procedure

Odds Ratio Estimates

Effect	episodes	Point Estimate	95% Wald Confidence Limits	
rank 2 vs 1	4	1.712	0.936	3.132
rank 2 vs 1	3	1.137	0.742	1.741
rank 2 vs 1	2	1.882	1.046	3.387
rank 3 vs 1	4	2.573	1.434	4.618
rank 3 vs 1	3	1.357	0.890	2.069
rank 3 vs 1	2	1.813	1.020	3.223
rank 4 vs 1	4	2.105	1.115	3.976
rank 4 vs 1	3	1.025	0.643	1.634
rank 4 vs 1	2	1.961	1.012	3.803
rank 5 vs 1	4	3.356	1.590	7.084
rank 5 vs 1	3	1.156	0.625	2.137
rank 5 vs 1	2	1.347	0.601	3.015
level 2 vs 1	4	1.142	0.765	1.705
level 2 vs 1	3	0.819	0.583	1.150
level 2 vs 1	2	0.964	0.609	1.528
level 3 vs 1	4	1.036	0.676	1.590
level 3 vs 1	3	1.109	0.774	1.589
level 3 vs 1	2	3.071	1.655	5.701

The only difference between this model than the partial is that now the level category is considered non-proportional. The interpretations remain the same. The Odds Ratio for the rank comparisons are the same as before and the level comparisons are for each response comparison. Inferences in this case become specific to individual cumulative dependent value comparisons.

CONCLUSION

The use of the UNEQUALSLOPES option is a powerful tool when the data doesn't cooperate in an even way among comparisons. The logistic procedure after the advent of SAS 9.3 makes it easier to create those odds ratios when the assumption of proportionality doesn't hold water. Being able to effectively communicate the results when the models get hairy is an important part of this. It is clear that although NLMIXED gives a bit more insight to what is going on, the logistic UNEQUALSLOPES model option makes the programming much easier and produces CI and p-values which better explain the relationships.

REFERENCES

High, Robin. (2012). *Models for Ordinal Response Data* Available at http://www.mwsug.org/proceedings/2012/SA/MWSUG-2012-SA09.pdf

Hosmer, David W Jr., Lemeshow, Stanley, and Sturdivant, Rodney X. (2013) *Applied Logistic Regression, Third Edition*, Hoboken, New Jersey: John Wiley & Sons

SAS® online support, "Usrank1 Note 22954: The PROC LOGISTIC proportional odds test and fitting a partial proportional odds model" available at http://support.sas.com/kb/22/954.html

ACKNOWLEDGMENTS

I'd like to thank Meihua Qiao and Karl Miller for their review.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Matthew Wiedel
inVentiv Health
Lincoln, NE 68516
Direct +1 402 480 5792
matthew.wiedel@inventivhealth.com